
ABSTRACT  
Multivariate statistical analysis plays an increasingly important role as the number of variables being 
measured increases in educational research. In both cognitive and noncognitive assessments, many 
instruments that researchers aim to study contain a large amount of variables where each measured 
variable can be assigned to a specific factor of a bigger construct. The factor of an instrument typically 
emerges through the use of educational theories or empirical research. Two types of factor analysis are 
widely used in order to understand the latent relationships among multiple variables based on different 
scenarios: (1) exploratory factor analysis (EFA), which is performed by using the SAS® PROC FACTOR 
procedure, is an advanced statistical method to probe deeply into the relationship among the variables 
and the larger construct and then develop a tailored model for the specific assessment; (2) confirmatory 
factor analysis (CFA), which is used after a model is established, is conducted by using the SAS® PROC 
CALIS procedure to examine the model fit of specific data and then make adjustment for the model as 
needed. This paper presents the application of SAS® to conduct these two types of factor analysis to 
fulfill various research purposes. Examples using noncognitive assessment data are demonstrated, and 
the interpretation of the fit statistics is discussed. 
 
 

INTRODUCTION  
Historically, cognitive skills have been believed to be the key factors for students’ academic achievement 
and life success (Brody, 1992; Bartels, Rietveld, Van Baal, & Boomsma, 2002). Over the last few 
decades, noncognitive skills have drawn more attention and have been found to be equally, if not more, 
important factors in predicting students’ academic achievement (Tracey & Sedlacek, 1982; Parker & 
Summerfeldt, 2004). In order to measure students’ noncognitive skills, a commonly used method is to ask 
students to complete a questionnaire to report their self-perceived feelings towards the items. The 
Mission Skills Assessment (MSA), used as our example, asks middle-school students to answer Likert-
type questions to measure six facets of noncognitive skills: teamwork, creativity, ethics, resilience, 
curiosity, and time management. 
 
When studying noncognitive skills, many validity and reliability concerns need to be considered for four 
reasons: (1) the nature of their original complexity; (2) responses are self-perceived; (3) the easiness to 
fake responses, and (4) the assessments are relatively low-stakes. Therefore, careful examinations to 
identify the measurement issues of noncognitive items before digging into the real research questions are 
of great significance. Common techniques include the following: (1) zero-variance checks to see if 
students paid attention while answering, (2) outlier checks for unreasonable response patterns, and (3) 
factor analysis to see if specified items measure the intended skill. The third method is the focus of this 
paper.  
 
This paper introduces concept of factor analysis, demonstrates main differences between EFA and CFA, 
and provides an example to illustrate implementation of these methods in SAS and interpretation of SAS 
output for research studies. 
 
 

FACTOR ANALYSIS 
Factor analysis is a multivariate statistical method that is implemented to detect the correlations between 
the variables observed and the potentially latent factors that are unobserved (Knott & Bartholomew, 
1999). It is commonly used in noncognitive assessments as researchers aim to measure students’ 
noncognitive attributes/skills without them being able to concretely measure those skills that are being 

detected by that specific question. Therefore, a series of carefully designed questions/prompts are given 
to the students to measure a variety of different noncognitive skills. Additionally, due to the fact that 
noncognitive skills are complex in nature, items sometimes measure unintended factors or the same 
assessment may produce different results when used for different groups of participants (Douglas, 
McDaniel & Snell, 1996). Therefore, factor analysis is a crucial and effective statistical method, 
functioning as the gate keeper before proceeding to more complex research questions, to ensure that the 
assessment is working correctly. 
 
Generally, there are two types of commonly used factor analysis: exploratory (EFA) and confirmatory 
(CFA) (Thompson, 2004). EFA is used to identify complex correlations among items and group items that 
are part of unified constructs. The researchers make no a priori assumptions about the relationships 
among factors. CFA, on the other hand, tests the hypothesis that the items are associated with specific 
factors. Hypothesized models are tested against actual data and the analysis would demonstrate loadings 
of observed variables on the latent variables, as well as the relationships between the latent variables. To 
summarize, researchers usually use EFA with the goal to establish a theory or a model, while they more 
often conduct CFA to validate a hypothesized model with actual data. 
 

 
EXAMPLES 
DATASET 
As previously mentioned, we use MSA data from a self-report Likert questionnaire administered in June 
2012 to 1,675 students from 14 middle schools across the United States. This questionnaire contains the 
two assessments discussed below. 
 
Motivation has been found to be a strong predictor of students’ academic outcomes (Wong & Mihaly, 
1991; Liu, Bridgeman & Adler, 2012). Based on psychological theories, researchers have divided 
motivation into four facets:  instrumental, parental, and competitive motivation, which are three types of 
extrinsic motivation, and intrinsic motivation. The questionnaire asked participants six questions to 
measure each of the four facets within different situations of motivation. Each question is asked on a four-
point response scale ranging from “Very Unlikely” to “Very Likely.” A full list of the twenty-four items is 
shown in Table A in the Appendix. We implement CFA by using the SAS PROC CALIS procedure to 
examine if the four-factor model holds for the 1,675 students. 
 
GRIT (equal to resilience in this case) has drawn increasing attention in the past decades. Duckworth 
(2007) invented GRIT. By her definition, GRIT is the resilience and passion that is crucial to achieve long-
term goals. In MSA data, GRIT was measured by 36 items on a four-point Likert response scale ranging 
from “Never or Rarely,” to “Usually or Always.” Twenty-two items were negative statements and the 
responses were reverse-coded for this analysis. A full list of the thirty-six items is shown in Table B in the 
Appendix. A one-factor CFA was originally conducted for these items and the model fit is far below the 
acceptance level, which indicates these thirty-six items are measuring multiple factors rather than one 
merged factor as designed. Therefore, an additional EFA is necessary to identify the latent factors behind 
the items. 
 
 
DIAGRAM OF CFA  
On the next page is the diagram of the proposed four-factor CFA model for the twenty-four motivation 
items. 

CODE OF CFA  
We use the following SAS code to conduct a four-factor CFA for the twenty-four motivation variables. The 
PROC CALIS procedure is used.  

proc calis data=msa; 

 lineqs 

  mot1  = p1  f_instru + e1, 

  mot2  = p2  f_instru + e2, 

  mot3  = p3  f_instru + e3, 

  mot4  = p4  f_instru + e4, 

  mot5  = p5  f_instru + e5, 

  mot6  = p6  f_instru + e6, 

  mot7  = p7  f_intrin + e7, 

  mot8  = p8  f_intrin + e8, 

  mot9  = p9  f_intrin + e9, 

  mot10 = p10 f_intrin + e10, 

  mot11 = p11 f_intrin + e11, 

  mot12 = p12 f_intrin + e12, 

  mot13 = p13 f_parent + e13, 

  mot14 = p14 f_parent + e14, 

  mot15 = p15 f_parent + e15, 

Motivation 

Instrumental 

Motivation 

mot1 

... 

... 

mot6 

Intrinsic 
Motivation 

mot7 

... 

... 

mot12 

Parental 
Motivation 

mot13 

... 

... 

mot18 

Competitive 
Motivation 

mot19 

... 

... 

mot24 

  mot16 = p16 f_parent + e16, 

  mot17 = p17 f_parent + e17, 

  mot18 = p18 f_parent + e18, 

  mot19 = p19 f_compet + e19, 

  mot20 = p20 f_compet + e20, 

  mot21 = p21 f_compet + e21, 

  mot22 = p22 f_compet + e22, 

  mot23 = p23 f_compet + e23, 

  mot24 = p24 f_compet + e24; 

 std 

  e1-e24 = vare1-vare24, 

  f_instru=1, 

  f_intrin=1, 

  f_parent=1, 

  f_compet=1; 

 cov 

  f_instru f_intrin = covf1f2, 

  f_instru f_parent = covf1f3, 

  f_instru f_compet = covf1f4, 

  f_intrin f_parent = covf2f3, 

  f_intrin f_compet = covf2f4, 

  f_parent f_compet = covf3f4; 

 var 

  mot1 mot2 mot3 mot4 mot5 mot6 

  mot7 mot8 mot9 mot10 mot11 mot12 

  mot13 mot14 mot15 mot16 mot17 mot18 

  mot19 mot20 mot21 mot22 mot23 mot24; 

run; 

 

RESULTS OF CFA 
By conducting the PROC CALIS procedure in SAS, we can see four sections in the output: (1) Model and 
Initial Values; (2) Descriptive Statistics; (3) Optimization; and (4) Maximum Likelihood Estimation. The Fit 
Summary table in the fourth section Maximum Likelihood Estimation contains the most important 
information regarding the model fit to our data. 
 

The CALIS Procedure 
Covariance Structure Analysis: Maximum Likelihood Estimation 

 

Fit Summary 

 …  

Absolute Index Fit Function 1.8286 

  Chi-Square 3061.1126 

  Chi-Square DF 246 

  Pr > Chi-Square <.0001 

  …  

 Parsimony Index RMSEA Estimate 0.0827 

  …  

Incremental Index Bentler Comparative Fit Index 0.8351 

  Bentler-Bonett NFI 0.8236 

  Bentler-Bonett Non-normed Index 0.8150 

  …  
 

Whether a model fit is satisfactory is determined by these parameters: the p value of the chi-square 
statistics, the Root Mean Square Error of Approximation (RMSEA), the comparative fit index (CFI) and 
Bentler-Bonett NFI. A satisfactory model fit is suggested by a nonsignificant chi square at a 0.05 level 
(Cheung & Rensvold, 2002), RMSEA values less than 0.08 (Browne & Cudeck, 1993), CFI values bigger 
than 0.95, and NFI greater than 0.90 (Byrne, 2006). However, due to the sensitive nature of the 
significance of chi square statistics, some literature suggested to exclude it from determining a model fit 
(Jöreskog, 1969). 
 
From the Fit Summary table, we can see the p-value of the chi-square probability is smaller than 0.0001, 
which indicates the chi-square statistic is significant at the 0.05 level. The RMSEA is 0.0827, which 
means the residual is slightly greater than the cutoff of 0.08. The CFI and NFI results are 0.84 and 0.82, 
with none of them meeting the criterion. To summarize, from the resulting indices, we can claim that the 
model fit is less than ideal and therefore a further examination regarding these 24 items in this specific 
dataset will be needed before any actual analysis moves forward. 
 
 

Standardized Results for Covariances Among Exogenous Variables 

Var1 Var2 Parameter Estimate Standard 
Error 

t Value 

f_instru f_intrin covf1f2 0.47489 0.02389 19.87480 

f_instru f_parent covf1f3 0.53200 0.02143 24.82774 

f_instru f_compet covf1f4 0.82413 0.01293 63.75380 

f_intrin f_parent covf2f3 0.10806 0.02899 3.72730 

f_intrin f_compet covf2f4 0.42669 0.02564 16.64316 

f_parent f_compet covf3f4 0.55098 0.02165 25.44929 

Additionally, we also examine the correlations among the four latent factors to see how they relate to 
each other. It is suggested to combine the factors if their correlations are extremely high. The covariances 
are presented above. The correlations are equivalent to the covariances in that the variances of the latent 
variables are equal to one. From the results table, we can find out that all correlations are small or 
medium, ranging from 0.10 to 0.60, expect for the correlation between instrumental motivation and 
competitive motivation, which is slighter higher than the cut off. Therefore, it is suggesting that 
instrumental motivation and competitive motivation are highly correlated and thus may need to be 
combined into one factor.  

 

DIAGRAM OF EFA 

Below is the diagram of the EFA model for the thirty-six GRIT variables. Unlike the CFA model, how many 
factors we have and which variable belongs to which factor is unknown and two consecutive EFAs will be 
used to determine these answers. The terminology will be assigned to each factor after the EFA analysis. 
In the diagram, resilience, self-efficacy, and emotional stability are three examples of GRIT that might 
evolve from the EFA. 

CODE OF EFA WITHOUT ROTATION 

proc factor data=msa method=ml scree priors=smc; 

  var grit1 grit2 grit3 grit4 grit5 grit6  

      grit7 grit8 grit9 grit10 grit11 grit12 

 grit13 grit14 grit15 grit16 grit17 grit18  

 grit19 grit20 grit21 grit22 grit23 grit24  

 grit25 grit26 grit27 grit28 grit29 grit30 

 grit31 grit32 grit33 grit34 grit35 grit36; 

run; 

 
RESULTS OF EFA WITHOUT ROTATION 

The FACTOR Procedure 
Initial Factor Method: Maximum Likelihood 

 
Preliminary Eigenvalues: Total = 23.8025853  

Average = 0.66118292 

  Eigenvalue Difference Proportion Cumulative 

1 15.2760100 10.6504948 0.6418 0.6418 

2 4.6255152 1.9186537 0.1943 0.8361 

3 2.7068614 1.2735814 0.1137 0.9498 

4 1.4332800 0.3989235 0.0602 1.0100 

5 1.0343565 0.3382835 0.0435 1.0535 

… ... … … … 

grit1 

Resilience 
grit2 

Self-Efficacy 

... 

Emotional 

Stability 

... 

grit35 

grit36 

The FACTOR Procedure 
Initial Factor Method: Maximum Likelihood 

 
Preliminary Eigenvalues: Total = 23.8025853  

Average = 0.66118292 

  Eigenvalue Difference Proportion Cumulative 

32 -0.3023499 0.0197613 -0.0127 1.0601 

33 -0.3221111 0.0284963 -0.0135 1.0465 

34 -0.3506074 0.0106259 -0.0147 1.0318 

35 -0.3612333 0.0349282 -0.0152 1.0166 

36 -0.3961616   -0.0166 1.0000 

4 factors will be retained by the PROPORTION criterion. 

 
From the output, we can see four factors were retained by the proportion criterion. The preliminary 
eigenvalues are 15.2760100, 4.6255152, 2.7068614, and 1.4332800. Each of the four factors explains 
64.18%, 19.43%, 11.37%, and 6.02% of the variance, respectively. 

Factor analysis has an advantage that the axes of the factors can be rotated within the multidimensional 
variable space. The rotation method is often used to make the results of the factor analysis more easily 
interpretable (Browne, 2001). With the knowledge that there are four factors retained in mind, we will then 
conduct another EFA with the rotation method by specifying the number of factors in the model. 

 
CODE OF EFA WITH FOUR FACTORS AND ROTATION 

proc factor data=msa method=ml rotate=v n=4 reorder plot out=facsub4     

priors=smc; 

  var grit1 grit2 grit3 grit4 grit5 grit6  

      grit7 grit8 grit9 grit10 grit11 grit12 

 grit13 grit14 grit15 grit16 grit17 grit18  

 grit19 grit20 grit21 grit22 grit23 grit24  

 grit25 grit26 grit27 grit28 grit29 grit30 

 grit31 grit32 grit33 grit34 grit35 grit36 

 ; 

RUN; 

In this code, the varimax rotation is selected and four factors will be kept in the result. 

 

RESULTS OF EFA WITH FOUR FACTORS AND ROTATION 

Rotated Factor Pattern 

  Factor1 Factor2 Factor3 Factor4 

grit3  0.71378 0.11327 0.16013 0.08056 

grit6  0.63243 0.13696 0.12069 -0.10226 

grit9  0.63195 0.15119 0.08723 0.01850 

grit1  0.58905 0.13210 0.08503 0.08759 

grit10  0.55832 0.09571 0.13220 0.03099 

grit2  0.55710 0.10377 0.16052 0.26539 

grit7  0.54499 0.11591 0.14978 -0.01258 

grit8  0.54431 0.06416 0.10458 0.06824 

grit11  0.51178 0.03730 0.15530 0.02681 

grit4  0.50774 -0.00863 0.11920 0.38322 

grit22  0.49070 0.02115 0.27041 0.45230 

grit5  0.47990 -0.02599 0.10864 0.21421 

grit18  0.47328 0.06216 0.31439 0.33591 

grit16  0.46778 0.19009 0.37246 -0.01841 

grit14  0.41361 0.08643 0.40227 0.08337 

grit32  0.21869 0.63444 0.07777 0.12400 

grit28  0.07460 0.62015 0.22659 0.07478 

grit27  0.14273 0.60679 0.03167 0.01584 

grit26  0.18648 0.59979 0.00943 0.01836 

grit29  0.10065 0.58177 0.40610 0.12202 

grit25  0.09222 0.53775 -0.03127 -0.01405 

grit31  0.10315 0.53210 0.09067 0.19550 

grit33  0.06264 0.51546 -0.01308 0.14538 

grit30  -0.01450 0.42668 -0.01673 0.00520 

grit24  -0.11449 0.24730 -0.05812 0.19524 

grit12  0.36884 0.06752 0.59511 0.15887 

grit17  0.37983 0.10135 0.59407 0.11124 

grit13  0.33741 0.02601 0.54371 0.24930 

Rotated Factor Pattern 

  Factor1 Factor2 Factor3 Factor4 

grit15  0.43628 0.11162 0.45303 0.07420 

grit19  0.22706 -0.04806 0.37826 0.56023 

grit34  -0.00256 0.30198 0.16246 0.53966 

grit23  0.11254 0.40145 0.01999 0.49850 

grit21  0.21337 -0.05383 0.43038 0.46632 

grit20  0.43035 -0.00920 0.24766 0.44959 

grit36  0.08005 0.37426 0.08522 0.41046 

grit35  -0.01953 0.24231 -0.02897 0.40609 

For the factor loadings, we use 0.40 as the cut off to decide if an item belongs to that factor. If one item 
has multiple loadings that are greater than 0.40, then it is a cross-loaded item. We will assign this item to 
the factor which has a bigger loading in our study, but it is recommended to further investigate the item as 
necessary. From the rotated factor pattern result, we can conclude that 15 items are correlated with factor 
1, 10 items are correlated with factor 2, four items are correlated with factor 3, and seven items are 
correlated with factor 4. It is important to note that five items are associated with more than one factor. 
Their loadings are underlined in the table and should be assigned to one of the two factors based on 
psychological or educational theory. From careful review of all the thirty-six items, we can distinguish that 
we have four distinct factors and assigning them names based on related psychological theories. In our 
case, factor 1 can be called resilience, factor 2 can be called self-efficacy, factor 3 can be called 
emotional stability, and factor 4 can be named conscientiousness. 

 

CONCLUSION 
Factor analysis is an important multivariate statistical analysis to disentangle the relationships between 
the items and their latent factors. Researchers usually use EFA in order to establish a theory or a model, 
or in our example, to explore the latent factors that actually exist in the data in search of better model fit. 
On the other hand, CFA should be applied to validate an existing model with collected data. The 
procedures outlined in this paper show that PROC CALIS and PROC FACTOR have the analytic 
capabilities to conveniently implement CFAs and EFAs respectively and produce thorough and 
informative results to guide future research. 

APPENDIX 

Table A. Details of the 24 Motivation Variables 

SAS Variable 
Name 

Item/Prompt Motivation Type 

mot1 
Studying hard is worth it because it will help me get into 
college one day. 

Instrumental 
Motivation 

mot2 Reading in my free time will help me to get into college. 

mot3 
It will give me practice mastering the mathematics skills that I 
will need to get into college. 

mot4 Paying attention in science will be useful for high school. 

mot5 
I study more for math class because I will learn many things in 
math that will help me get into the best college. 

mot6 
Good grades in social studies will help me get into the college 
I want. 

mot7 
Studying for quizzes is enjoyable because I learn while I am 
doing it. 

Intrinsic 
Motivation 

mot8 Reading in my free time is enjoyable. 

mot9 I am interested in learning how to solve math problems. 

mot10 I pay attention in science class because I enjoy it. 

mot11 I study more for math class because math is fun. 

mot12 I enjoy my social studies class. 

mot13 
I will not go out because my parents expect me to do well on 
the quiz. 

Parental 
Motivation 

mot14 My parents want me to read as much as I can. 

mot15 My parents encouraged me to enroll in the club. 

mot16 
I pay attention in science because my parents will be. 
disappointed if I do not do well in science. 

mot17 I study more for math class because my parents want me to. 

mot18 My parents expect me to do well in social studies. 

mot19 
It is more important for to me to get good grades than to go 
out with my friends. 

Competitive 
Motivation 

mot20 
Reading in my free time will help me get better grades than 
the other students in my class. 

mot21 
I want to obtain the highest grade in my math classes and 
joining this club will help. 

mot22 
Paying attention in science will help me do better on exams 
than the other students in my class. 

mot23 
I study more for math class because I want to get higher 
grades than anyone else. 

mot24 I want to get better grades than the other students in my class. 

Table B. Details of the 36 GRIT Variables 

SAS Variable 
Name 

Item/Prompt 

grit1 I give up easily when faced with an obstacle. (reversed) 

grit2 I am easily discouraged. (reversed) 

grit3 I give up easily. (reversed) 

grit4 I am easily frustrated. (reversed) 

grit5 I get annoyed by people. (reversed) 

grit6 I avoid responsibilities. (reversed) 

grit7 I forget to do things. (reversed) 

grit8 I make careless mistakes. (reversed) 

grit9 I quickly lose interest in the tasks I start. (reversed) 

grit10 I react slowly. (reversed) 

grit11 My interests change quickly. (reversed) 

grit12 Sometimes I feel depressed. (reversed) 

grit13 Sometimes when I fail I feel worthless. (reversed) 

grit14 Sometimes, I do not feel in control of my school work. (reversed) 

grit15 I am filled with doubts about my competence. (reversed) 

grit16 I do not feel in control of my success in school. (reversed) 

grit17 There are times when things look pretty bleak and hopeless to me. (reversed) 

grit18 I get discouraged when things go wrong. (reversed) 

grit19 I worry. (reversed) 

grit20 I get stressed out easily when things don’t go my way. (reversed) 

grit21 I worry about school. (reversed) 

grit22 I get upset easily. (reversed) 

grit23 I remain calm under pressure. 

grit24 I seldom get mad. 

grit25 I am diligent. 

grit26 I complete tasks successfully. 

grit27 When I try, I generally succeed. 

grit28 I am confident I get the success I deserve in life. 

grit29 Overall, I am satisfied with myself. 

grit30 I determine what will happen in my life. 

grit31 I am capable of coping with most of my problems. 

grit32 I overcome challenges and setbacks. 

grit33 I am resilient. 

grit34 I am relaxed. 

grit35 I am not easily bothered by things. 

grit36 I remain calm when I have a lot of homework to do. 

 


